
1. [bookmark: _Toc122434485][bookmark: _Toc125125650]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.3.6 which is part of the LTE_A_PRO test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc125125651]Verification Test Summary 
Test Case: 	11.3.6
ATS Version:	iwd-TTCN3-B2022-09_D22wk49
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	Qualcomm SA515M
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc125125652]Corrections required
1. [bookmark: _Toc125125653][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	f_TC_11_3_6_EUTRA()

	Reason for change
	The UE may (or may not) have performed detach procedure during test body   

	Summary of change
	Moved postamble handling into test body function (see change 2)


	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	



Before change
  function f_TC_11_3_6_EUTRA() runs on EUTRA_PTC
  { /* eCall Only mode / Limited service state / Call to URI for test service should not be attempted / eCall over IMS should be attempted */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    //Set PLMN identifier to PLMN4
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_CellA, cs_HPLMN_004_31);

    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    // The UE is equipped with a USIM containing values shown in Table 11.3.6.3.1-2.
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2 and contains the values shown in Table 11.3.6.3.1-2");

    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_6_Body();
    
    // Postamble
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E1_IDLE); // optional IMS Deregistration is handled in the default, before the UE detaches
  }

    

After change
  function f_TC_11_3_6_EUTRA() runs on EUTRA_PTC
  { /* eCall Only mode / Limited service state / Call to URI for test service should not be attempted / eCall over IMS should be attempted */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    //Set PLMN identifier to PLMN4
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_CellA, cs_HPLMN_004_31);

    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    // The UE is equipped with a USIM containing values shown in Table 11.3.6.3.1-2.
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2 and contains the values shown in Table 11.3.6.3.1-2");

    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_6_Body();
    
    // Postamble
    // MOVED f_EUTRA_TestBody_Set(false);

    // MOVED f_EUTRA_Postamble(eutra_CellA, E1_IDLE); // optional IMS Deregistration is handled in the default, before the UE detaches

    // Test ends in fl_TC_11_3_6_Body
  }
   

1. [bookmark: _Toc125125654]Change 2
	Function name
	fl_TC_11_3_6_Body()

	Reason for change
	1. At the beginning of the test case IP handling should be started for the emergency PDN (PDN_2 using DRB 2)

2. Incorrect EPS bearer ID used for deactivation
3. In this Limited Service scneario the UE will attach for EPS emergency services, so the attach type used is “EPS emergency attach”. However, during the postamble the detach type evaluated from PIXIT settings may be “combined EPS/IMSI detach”.
4. DETACH REQUEST message at step 27Aa1 will be received on SRB2 not SRB1
5. The t_Watchdog timer is not stopped
6. The UE may perform detach at step 27Aa2 

	Summary of change
	1. Pass PDN_2 and DRB 2 into  f_EUTRA_SwitchOnUEandStartIP()

2. Correct bearer ID to tsc_EpsDedicatedBearerId2 in call to    f_EUTRA_DeactivateEPS_BearerContext()




3. Pass p_ForcedAttach  “EPS_Only” in postamble function. 
4. Replace SRB1 with SRB2 for receiving DETACH REQUEST and sending DETACH ACCEPT.
5. Stop t_Watchdog on receipt of DETACH REQUEST
6. Add postamble call in this function, with state depending on if the UE already detached during the test body.


	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	



Before change
  function fl_TC_11_3_6_Body() runs on EUTRA_PTC
  {
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_MSG_Indication_Type v_NasInd;
    var NAS_AttDetValue_Type v_DetachType;
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var float v_EcallDeregisteredInactive := 60.0;
    timer t_EcallNoRRCConnReq := 120.0;
    timer t_Watchdog := 5.0;

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated.
    f_UT_InitiateECall(UT, TEST);

    //@siclog "Step 4" siclog@
    // Check: Does UE send an RRCConnectionRequest on Cell A within 120 seconds?
    t_EcallNoRRCConnReq.start;
    alt {
      [] t_EcallNoRRCConnReq.timeout {} // Continue the test
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - Unexpected RRCConnectionRequest");
        }
      }

    //@siclog "Step 5" siclog@
    // A manual eCall is initiated.
    f_UT_InitiateECall(UT, MANUAL);

    //@siclog "Step 6" siclog@
    // Check: Does the UE transmit an RRCConnectionRequest message with 'establishmentCause' set to 'emergency'?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, emergency);

    //@siclog "Step 7" siclog@
    // SS transmits an RRCConnectionSetup message.
    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, -, tsc_RRC_TI_Def);

    //@siclog "Step 8" siclog@
    // Check: Does the UE transmit an ATTACH REQUEST message to attach for emergency bearer services, together with a PDN CONNECTIVITY REQUEST message with 'Request type' set to 'emergency?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                          cr_508_ATTACH_REQUEST(tsc_EAttachType_Emergency,
                                                                                                                v_AdditionalUpdateType),
                                                                                          cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,
                                                                                                                          f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    //@siclog "Step 9-18" siclog@
    // Steps 5 to 14 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps7_16(eutra_CellA, v_NasInd); // @sic R5-227236 sic@

    //EXCEPTION: In parallel to the events described in steps 19 to 23 below, the behaviour in table 11.3.6.2-2 occurs.

    //@siclog "Step 19-23" siclog@
    // Steps 15 to 19 of the generic test procedure for IMS emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19(eutra_CellA); // @sic R5-227236 sic@

    //@siclog "Step 24-27" siclog@
    // Release eCall over IMS using the generic procedure described in TS 34.229-1 [35] subclause C.33  @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId); // there is only one PDN

    //EXCEPTION: Step 27Aa1 and 27Aa2 describes optional behaviour that depends on the UE implementation.
    v_DetachType :=  f_GetEDetachType(NORMAL);
    t_Watchdog.start;
    alt {
      [] t_Watchdog.timeout {} // Continue the test
      [] //@siclog "Step 27Aa1" siclog@
         // The UE transmits a DETACH REQUEST message with the Detach type IE indicating "EPS detach" to regain normal service.
         SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA, //@sic R5-227236 sic@
                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                           cr_508_DETACH_REQUEST_MO(?,
                                                                                    f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                    cr_DetachType( '0'B, v_DetachType))))) {
         //@siclog "Step 27Aa2" siclog@
         // The SS responds the DETACH ACCEPT message.
         SRB.send (cas_SRB1_NasPdu_REQ (eutra_CellA, //@sic R5-227236 sic@
                                        cs_TimingInfo_Now,
                                        cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                       cs_508_DETACH_ACCEPT))); };
      }

    //@siclog "Step 29" siclog@
    // The SS releases the RRC connectionafter 5secs of Step 27..
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
  }


After change
 function fl_TC_11_3_6_Body() runs on EUTRA_PTC
  {
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_MSG_Indication_Type v_NasInd;
    var NAS_AttDetValue_Type v_DetachType;
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var float v_EcallDeregisteredInactive := 60.0;
    timer t_EcallNoRRCConnReq := 120.0;
    timer t_Watchdog := 5.0;
    var boolean v_Detached := false;

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA, -, -, -, PDN_2, tsc_DRB2);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated.
    f_UT_InitiateECall(UT, TEST);

    //@siclog "Step 4" siclog@
    // Check: Does UE send an RRCConnectionRequest on Cell A within 120 seconds?
    t_EcallNoRRCConnReq.start;
    alt {
      [] t_EcallNoRRCConnReq.timeout {} // Continue the test
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - Unexpected RRCConnectionRequest");
        }
      }

    //@siclog "Step 5" siclog@
    // A manual eCall is initiated.
    f_UT_InitiateECall(UT, MANUAL);

    //@siclog "Step 6" siclog@
    // Check: Does the UE transmit an RRCConnectionRequest message with 'establishmentCause' set to 'emergency'?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, emergency);

    //@siclog "Step 7" siclog@
    // SS transmits an RRCConnectionSetup message.
    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, -, tsc_RRC_TI_Def);

    //@siclog "Step 8" siclog@
    // Check: Does the UE transmit an ATTACH REQUEST message to attach for emergency bearer services, together with a PDN CONNECTIVITY REQUEST message with 'Request type' set to 'emergency?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                          cr_508_ATTACH_REQUEST(tsc_EAttachType_Emergency,
                                                                                                                v_AdditionalUpdateType),
                                                                                          cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,
                                                                                                                          f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    //@siclog "Step 9-18" siclog@
    // Steps 5 to 14 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps7_16(eutra_CellA, v_NasInd); // @sic R5-227236 sic@

    //EXCEPTION: In parallel to the events described in steps 19 to 23 below, the behaviour in table 11.3.6.2-2 occurs.

    //@siclog "Step 19-23" siclog@
    // Steps 15 to 19 of the generic test procedure for IMS emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19(eutra_CellA); // @sic R5-227236 sic@

    //@siclog "Step 24-27" siclog@
    // Release eCall over IMS using the generic procedure described in TS 34.229-1 [35] subclause C.33  @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                                      tsc_EpsDedicatedBearerId2); // there is only one PDN

    //EXCEPTION: Step 27Aa1 and 27Aa2 describes optional behaviour that depends on the UE implementation.
    
    v_DetachType :=  f_GetEDetachType(EPS_Only);
    t_Watchdog.start;
    alt {
      [] t_Watchdog.timeout {} // Continue the test
      [] //@siclog "Step 27Aa1" siclog@
         // The UE transmits a DETACH REQUEST message with the Detach type IE indicating "EPS detach" to regain normal service.
             SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, //@sic R5-227236 sic@
                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                           cr_508_DETACH_REQUEST_MO(?,
                                                                                    f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                    cr_DetachType( '0'B, v_DetachType))))) {
         t_Watchdog.stop;
         //@siclog "Step 27Aa2" siclog@
         // The SS responds the DETACH ACCEPT message.
         SRB.send (cas_SRB2_NasPdu_REQ (eutra_CellA, //@sic R5-227236 sic@
                                        cs_TimingInfo_Now,
                                        cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                       cs_508_DETACH_ACCEPT)));
         v_Detached := true;
       }
     }

    //@siclog "Step 29" siclog@
    // The SS releases the RRC connectionafter 5secs of Step 27..
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
    f_EUTRA_TestBody_Set(false);

    if( v_Detached == false) {  
      f_EUTRA_Postamble(eutra_CellA, E1_IDLE, EPS_Only);
    }
    else {
      f_EUTRA_Postamble(eutra_CellA, E4_DEREGISTERED);
    }
    
}

 
   

1. [bookmark: _Toc125125655]Change 3
	Function name
	f_TC_11_3_6_IMS2()

	Reason for change
	1. Invite request would be received with context A6 (INVITE for creating an emergency session in case of no registration) as per 34.229-1 and section A.2.1, see change 2.4
2. Following emergency call release, associated TCP sockets should be closed before  IPCAN release.

	Summary of change
	1. Removed emergency registration procedure
2. Added function to close TCP sockets.

	TTCN module
	 IMS_ECall_IMS.ttcn

	MCC160 Comment
	



Before change
  function f_TC_11_3_6_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 */
    var SipUrl v_ContactUrl;
    
    f_IMS_PTC_Init(PDN_2);

    f_IMS_EmergencyRegistration();
   
    f_ECall_CallSetup_AnnexC47(MANUAL);
    f_ECall_MSDUpdate_AnnexC47();
    
    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrlUE(dialog);
    f_IMS_CallReleaseMT(v_ContactUrl);

    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistration();
  }


After change
function f_TC_11_3_6_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 */
    var SipUrl v_ContactUrl;
    
    f_IMS_PTC_Init(PDN_2);

    // REMOVED f_IMS_EmergencyRegistration(); 
   
    f_ECall_CallSetup_AnnexC47(MANUAL);
    f_ECall_MSDUpdate_AnnexC47();
    
    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrlUE(dialog);
    f_IMS_CallReleaseMT(v_ContactUrl);

    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_TCP_CloseHandling(waitAndClose);

    f_IMS_IpcanReleaseWithOptionalDeregistration();
  }

1. [bookmark: _Toc125125656]Change 4
	Function name
	f_ECall_InitialInvite_AnnexC47()

	Reason for change
	According to note 1 in Table 11.3.6.3.2-2: Parallel behaviour, Invite request would be received with context A6 as per 34.229-1 and section A.2.1 


	Summary of change
	Passed the context as A_2_1_A6 as input to the Annex C47 procedure.

	TTCN module
	IMS_eCall_Functions.ttcn

	MCC160 Comment
	



Before change
function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21, A25 @sic R5-227567 sic@
    v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-227567, R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }




After change
function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    if(testcasename()== "TC_11_3_6") {
      v_Context := A_2_1_A6;
    }
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21, A25 @sic R5-227567 sic@
    v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-227567, R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }



1. [bookmark: _Toc125125657]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on EUTRA band 4 using ciphering algorithm eea1 and integrity algorithm eia1 
3 [bookmark: _Toc125125658]Execution Log Files
3.1 [bookmark: _Toc125125659][bookmark: _Hlk120026754]Qualcomm SA515M
The Qualcomm SA515M passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_11_3_6_LOG.html
· PICS/PIXIT parameter file:
       TC_11_3_6_PIXIT.xml


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc125125660]References 
	[1]
	R5s230116 : Supporting information for  addition of eCall over IMS test case 11.3.6



